Cálculo 3 - 2024.2

P1 (primeira prova)

Eduardo Ochs - RCN/PURO/UFF
http://anggtwu.net/2024.2-C3.html

Links

http://anggtwu.net/e/maxima.e.html#2024.2-C3-P1-Q1
http://anggtwu.net/e/maxima.e.html#2024.2-C3-P1-Q2

```
(find-es "maxima" "2024.2-C3-P1-Q1")
(find-es "maxima" "2024.2-C3-P1-Q2")
```

Questão 1

(Total: 3.5 pts)
O diagrama de numerozinhos da última folha da prova corresponde a uma superfície $z = F(x,y)$ que tem 6 faces. Também é possível interpretá-lo como uma superfície com 7 ou mais faces, mas vamos considerar que a superfície com só 6 faces é que é a correta.

a) **(0.5 pts)** Mostre como dividir o plano em 6 polígonos que são as projeções destas faces no plano do papel.

b) **(0.5 pts)** Chame estas faces de face N ("norte"), S ("sul"), W ("oeste"), C ("centro"), E ("leste") e NE ("nordeste"), e chame as equações dos planos delas de $F_N(x,y)$, $F_S(x,y)$, $F_W(x,y)$, $F_C(x,y)$, $F_E(x,y)$, e $F_{NE}(x,y)$. Dê as equações destes planos.

c) **(0.5 pts)** Sejam:

$$\begin{aligned} P_C &= \{ (x,y,z) \in \mathbb{R}^3 \mid z = F_C(x,y) \}, \\ P_E &= \{ (x,y,z) \in \mathbb{R}^3 \mid z = F_E(x,y) \}, \\ r &= P_C \cap P_E. \end{aligned}$$

Represente a reta r graficamente como numerozinhos.

d) **(0.5 pts)** Dê uma parametrização para a reta do item anterior. Use notação de conjuntos.

e) **(0.5 pts)** Seja

$$A = \{0, 1, \ldots, 9\} \times \{0, 1, \ldots, 11\};$$

note que os numerozinhos do diagrama de numerozinhos estão todos sobre pontos de A. Para cada ponto $(x,y) \in A$ represente graficamente $(x,y) + \frac{1}{3}\vec{\nabla} F(x,y)$.

Obs: quando $\vec{\nabla} F(x,y) = 0$ desenhe uma bolinha preta sobre o ponto (x,y), e quando $\vec{\nabla} F(x,y)$ não existir faça um '×' sobre o numerozinho que está no ponto (x,y).

f) **(1.0 pts)** Sejam

$$\begin{aligned} Q(t) &= (0,2) + t\overrightarrow{(1,1)}, \\ (x(t), y(t)) &= Q(t), \\ h(t) &= F(x(t), y(t)). \end{aligned}$$

Faça o gráfico da função $h(t)$. Considere que o domínio dela é o intervalo $[0, 9]$.

Algumas definições

Em Cálculo 1 e Cálculo 2 você viu que se $f(x)$ é uma função de $\mathbb{R}$ em $\mathbb{R}$ então a aproximação de Taylor de ordem 2 pra $f(x)$ no ponto x_0 é:

$$\begin{array}{rcl}(T_{2,x_0}f)(x) & = & f(x_0) \\ & + & f'(x_0)\Delta x \\ & + & \frac{f''(x_0)}{2}\Delta x^2\end{array}$$

A “versão Cálculo 3” disto é a fórmula abaixo. Se $F(x,y)$ é uma função de $\mathbb{R}^2$ em $\mathbb{R}$ então a aproximação de Taylor de ordem 2 pra $F(x,y)$ no ponto (x_0,y_0) é:

$$\begin{array}{rcl}(T_{2,(x_0,y_0)}F)(x) & = & F(x_0,y_0) \\ & + & F_x(x_0,y_0)\Delta x + F_y(x_0,y_0)\Delta y \\ & + & \frac{F_{xx}(x_0,y_0)}{2}\Delta x^2 + F_{xy}(x_0,y_0)\Delta x\Delta y + \frac{F_{yy}(x_0,y_0)}{2}\Delta y^2\end{array}$$

e a gente diz que as derivadas até ordem 2 da função F são as funções $(F, F_x, F_y, F_{xx}, F_{xy}, F_{yy})$. Eu costumo organizar elas numa matriz:

$$D_2F = \begin{pmatrix} F & & \\ F_x & F_y & \\ F_{xx} & F_{xy} & F_{yy} \end{pmatrix}$$

$$(D_2F)(x_0,y_0) = \begin{pmatrix} F(x_0,y_0) & & \\ F_x(x_0,y_0) & F_y(x_0,y_0) & \\ F_{xx}(x_0,y_0) & F_{xy}(x_0,y_0) & F_{yy}(x_0,y_0) \end{pmatrix}$$

Questão 2

(Total: 6.5 pts)
Sejam

$$\begin{aligned} F(x,y) &= xy(6-2x-y), \\ P_1 &= (0,6), \\ P_2 &= (1,2), \\ P_3 &= (3,0), \\ P_4 &= (0,0). \end{aligned}$$

a) **(0.5 pts)** Calcule D_2F.

b) **(0.5 pts)** Calcule D_2F nos pontos P_1, P_2, P_3, e P_4.

c) **(1.0 pts)** Calcule $T_{2,(x_0,y_0)}F$ nos pontos P_1, P_2, P_3, e P_4.

d) **(0.5 pts)** Os pontos P_1, P_2, P_3 e P_4 são pontos críticos da função F? Quais deles são máximos locais? Quais são mínimos locais? Quais são pontos de sela? Use o gradiente e o determinante $\begin{vmatrix} F_{xx} & F_{xy} \\ F_{yx} & F_{yy} \end{vmatrix}$ pra descobrir tudo isso.

Lembre que $P_2 = (1,2)$.
Seja $G(x,y) = (T_{2,(1,2)}F)(x,y)$.
Seja $B = \{0,...,3\} \times \{0,...,6\}$
e $C = \{\,(x,y) \in B \mid y \le 6-2x\,\}$.

e) **(0.5 pts)** Calcule o diagrama de numerozinhos da função F nos pontos de C.

f) **(1.0 pts)** Calcule o diagrama de numerozinhos da função G nos pontos de C.

g) **(2.5 pts)** Use o diagrama de numerozinhos da F que você calculou no item (e) e os gradientes da F nos pontos de C – que você ainda não calculou, e vai ter que calcular agora – pra fazer um desenho bem caprichado das curvas de nível da F dentro do triângulo cujos vértices são os pontos P_1, P_3 e P_4. Você vai precisar reduzir a escala dos vetores gradientes pra que eles não esbarrem uns nos outros – desenhe $F(x,y) + \frac{1}{10}\nabla F(x,y)$ para cada ponto de C.

6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	5	5	5	5	5
6	6	6	6	5	4	4	4	4	4
6	6	6	5	4	3	2	2	2	2
5	5	5	4	3	2	1	0	0	0
4	4	4	3	2	1	0	0	0	0
3	3	3	2	1	0	0	0	0	0
2	2	2	1	0	0	0	0	0	0
1	1	1	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0

6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	5	5	5	5	5
6	6	6	6	5	4	4	4	4	4
6	6	6	5	4	3	2	2	2	2
5	5	5	4	3	2	1	0	0	0
4	4	4	3	2	1	0	0	0	0
3	3	3	2	1	0	0	0	0	0
2	2	2	1	0	0	0	0	0	0
1	1	1	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0

6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	5	5	5	5	5
6	6	6	6	5	4	4	4	4	4
6	6	6	5	4	3	2	2	2	2
5	5	5	4	3	2	1	0	0	0
4	4	4	3	2	1	0	0	0	0
3	3	3	2	1	0	0	0	0	0
2	2	2	1	0	0	0	0	0	0
1	1	1	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0

6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	5	5	5	5	5
6	6	6	6	5	4	4	4	4	4
6	6	6	5	4	3	2	2	2	2
5	5	5	4	3	2	1	0	0	0
4	4	4	3	2	1	0	0	0	0
3	3	3	2	1	0	0	0	0	0
2	2	2	1	0	0	0	0	0	0
1	1	1	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0

6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	5	5	5	5	5
6	6	6	6	5	4	4	4	4	4
6	6	6	5	4	3	2	2	2	2
5	5	5	4	3	2	1	0	0	0
4	4	4	3	2	1	0	0	0	0
3	3	3	2	1	0	0	0	0	0
2	2	2	1	0	0	0	0	0	0
1	1	1	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0

6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	6	6	6	6	6
6	6	6	6	6	5	5	5	5	5
6	6	6	6	5	4	4	4	4	4
6	6	6	5	4	3	2	2	2	2
5	5	5	4	3	2	1	0	0	0
4	4	4	3	2	1	0	0	0	0
3	3	3	2	1	0	0	0	0	0
2	2	2	1	0	0	0	0	0	0
1	1	1	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0

Questão 1: gabarito



Questão 1: gabarito (2)

```
(%i1) mkmatrix5(x,xs,y,ys,expr) ::=
        buildq([x,xs,y,ys,expr],
               apply('matrix,
                     makelist(makelist(expr,x,xs),y,ys)))$
(%i2) /* (1a) */
      /* (1b) */
      z_N  : 6$
(%i3) z_S  : 0$
(%i4) z_W  : y - 1;
```

(%o4)

$$y-1$$

```
(%i5) z_C  : y - x + 1;
```

(%o5)

$$y-x+1$$

```
(%i6) z_E  : -12 + 2*y;
```

(%o6)

$$2\,y-12$$

```
(%i7) z_NE : -4 + y;
```

(%o7)

$$y-4$$

```
(%i8) z_MR : min(z_E, z_NE);            /* middle right */
```

(%o8)

$$\min\left(y-4,2\,y-12\right)$$

```
(%i9) z_M  : min(z_W, max(z_C, z_MR));  /* middle */
```

(%o9)

$$\min\left(\max\left(\min\left(y-4,2\,y-12\right),y-x+1\right),y-1\right)$$

```
(%i10) z    : min(z_N, max(z_S, z_M))$
```

```
(%i11) mkmatrix5(x,seq(0,9), y,seqby(11,0,-1), [x,y]);
```

(%o11)

$$\begin{pmatrix}
[0,11] & [1,11] & [2,11] & [3,11] & [4,11] & [5,11] & [6,11] & [7,11] & [8,11] & [9,11] \\
[0,10] & [1,10] & [2,10] & [3,10] & [4,10] & [5,10] & [6,10] & [7,10] & [8,10] & [9,10] \\
[0,9] & [1,9] & [2,9] & [3,9] & [4,9] & [5,9] & [6,9] & [7,9] & [8,9] & [9,9] \\
[0,8] & [1,8] & [2,8] & [3,8] & [4,8] & [5,8] & [6,8] & [7,8] & [8,8] & [9,8] \\
[0,7] & [1,7] & [2,7] & [3,7] & [4,7] & [5,7] & [6,7] & [7,7] & [8,7] & [9,7] \\
[0,6] & [1,6] & [2,6] & [3,6] & [4,6] & [5,6] & [6,6] & [7,6] & [8,6] & [9,6] \\
[0,5] & [1,5] & [2,5] & [3,5] & [4,5] & [5,5] & [6,5] & [7,5] & [8,5] & [9,5] \\
[0,4] & [1,4] & [2,4] & [3,4] & [4,4] & [5,4] & [6,4] & [7,4] & [8,4] & [9,4] \\
[0,3] & [1,3] & [2,3] & [3,3] & [4,3] & [5,3] & [6,3] & [7,3] & [8,3] & [9,3] \\
[0,2] & [1,2] & [2,2] & [3,2] & [4,2] & [5,2] & [6,2] & [7,2] & [8,2] & [9,2] \\
[0,1] & [1,1] & [2,1] & [3,1] & [4,1] & [5,1] & [6,1] & [7,1] & [8,1] & [9,1] \\
[0,0] & [1,0] & [2,0] & [3,0] & [4,0] & [5,0] & [6,0] & [7,0] & [8,0] & [9,0]
\end{pmatrix}$$

```
(%i12) mkmatrix5(x,seq(0,8), y,seqby(11,0,-1), ''z);
```

(%o12)

$$\begin{pmatrix}
6 & 6 & 6 & 6 & 6 & 6 & 6 & 6 & 6 \\
6 & 6 & 6 & 6 & 6 & 6 & 6 & 6 & 6 \\
6 & 6 & 6 & 6 & 6 & 5 & 5 & 5 & 5 \\
6 & 6 & 6 & 6 & 5 & 4 & 4 & 4 & 4 \\
6 & 6 & 6 & 5 & 4 & 3 & 2 & 2 & 2 \\
5 & 5 & 5 & 4 & 3 & 2 & 1 & 0 & 0 \\
4 & 4 & 4 & 3 & 2 & 1 & 0 & 0 & 0 \\
3 & 3 & 3 & 2 & 1 & 0 & 0 & 0 & 0 \\
2 & 2 & 2 & 1 & 0 & 0 & 0 & 0 & 0 \\
1 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0
\end{pmatrix}$$

```
(%i13) /*
       plot3d (z,   [x,0,8],   [y,0,11]);
       */
```

Questão 1: gabarito (3)

```
(%i13) /* (1c) */
            [zr_=z_C, zr_=z_E];
(%o13)
```

$$[\mathrm{zr\_}=y-x+1,\mathrm{zr\_}=2\,y-12]$$

```
(%i14)     solve([zr_=z_C, zr_=z_E], [y,zr_]);
(%o14)
```

$$[[y=13-x,\mathrm{zr\_}=14-2\,x]]$$

```
(%i15) eqc : solve([zr_=z_C, zr_=z_E], [y,zr_])[1];
(%o15)
```

$$[y=13-x,\mathrm{zr\_}=14-2\,x]$$

```
(%i16) define(yr_(x), subst(eqc, y));
(%o16)
```

$$\mathrm{yr\_}(x):=13-x$$

```
(%i17) define(zr_(x), subst(eqc, zr_));
(%o17)
```

$$\mathrm{zr\_}(x):=14-2\,x$$

```
(%i18) xyzr(x)    := [x, yr_(x), zr_(x)];
(%o18)
```

$$\mathrm{xyzr}(x):=[x,\mathrm{yr\_}(x),\mathrm{zr\_}(x)]$$

```
(%i19) xyzr_top   : rhs(fundef(xyzr));
(%o19)
```

$$[x,\mathrm{yr\_}(x),\mathrm{zr\_}(x)]$$

```
(%i20) xyzr_lines : makelist(xyzr(x), x,2,9);
(%o20)
```

$$[[2,11,10],[3,10,8],[4,9,6],[5,8,4],[6,7,2],[7,6,0],[8,5,-2],[9,4,-4]]$$

```
(%i21) apply('matrix, append([xyzr_top], xyzr_lines));
(%o21)
```

$$\begin{pmatrix} x & \mathrm{yr\_}(x) & \mathrm{zr\_}(x) \\ 2 & 11 & 10 \\ 3 & 10 & 8 \\ 4 & 9 & 6 \\ 5 & 8 & 4 \\ 6 & 7 & 2 \\ 7 & 6 & 0 \\ 8 & 5 & -2 \\ 9 & 4 & -4 \end{pmatrix}$$

```
(%i22)  /* (1d) */
     [x, yr_(x), zr_(x)];
(%o22)
```

$$[x,13-x,14-2\,x]$$

Questão 1: gabarito (4)

```
(%i23) /* (1e) */
       define(z(x,y), z);
```
(%o23)
$$z\left(x,y\right) := \min\left(6, \max\left(0, \min\left(\max\left(\min\left(y-4, 2\,y-12\right), y-x+1\right), y-1\right)\right)\right)$$

```
(%i24) eps : 1/4;
```
(%o24)
$$\frac{1}{4}$$

```
(%i25) z_xr (x,y) := (z(x+eps,y)-z(x,y))/ eps;
```
(%o25)
$$\mathrm{z\_xr}\left(x,y\right) := \frac{z\left(x+\mathrm{eps},y\right)-z\left(x,y\right)}{\mathrm{eps}}$$

```
(%i26) z_xl (x,y) := (z(x-eps,y)-z(x,y))/-eps;
```
(%o26)
$$\mathrm{z\_xl}\left(x,y\right) := \frac{z\left(x-\mathrm{eps},y\right)-z\left(x,y\right)}{-\mathrm{eps}}$$

```
(%i27) z_yu (x,y) := (z(x,y+eps)-z(x,y))/ eps;
```
(%o27)
$$\mathrm{z\_yu}\left(x,y\right) := \frac{z\left(x,y+\mathrm{eps}\right)-z\left(x,y\right)}{\mathrm{eps}}$$

```
(%i28) z_yd (x,y) := (z(x,y-eps)-z(x,y))/-eps;
```
(%o28)
$$\mathrm{z\_yd}\left(x,y\right) := \frac{z\left(x,y-\mathrm{eps}\right)-z\left(x,y\right)}{-\mathrm{eps}}$$

```
(%i29) gradz(x,y) := if (z_xr(x,y) = z_xl(x,y)) and
                         (z_yu(x,y) = z_yd(x,y))
                      then [z_xr(x,y), z_yu(x,y)]
                      else "X"$
```

```
(%i30) mkmatrix5(x,seq(0,8), y,seqby(11,0,-1), gradz(x,y));
```
(%o30)
$$\begin{pmatrix}
[0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] \\
[0,0] & [0,0] & [0,0] & [0,0] & [0,0] & \mathrm{X} & \mathrm{X} & \mathrm{X} & \mathrm{X} \\
[0,0] & [0,0] & [0,0] & [0,0] & \mathrm{X} & \mathrm{X} & [0,1] & [0,1] & [0,1] \\
[0,0] & [0,0] & [0,0] & \mathrm{X} & [-1,1] & \mathrm{X} & \mathrm{X} & \mathrm{X} & \mathrm{X} \\
\mathrm{X} & \mathrm{X} & \mathrm{X} & [-1,1] & [-1,1] & [-1,1] & \mathrm{X} & [0,2] & [0,2] \\
[0,1] & [0,1] & \mathrm{X} & [-1,1] & [-1,1] & [-1,1] & [-1,1] & \mathrm{X} & \mathrm{X} \\
[0,1] & [0,1] & \mathrm{X} & [-1,1] & [-1,1] & [-1,1] & \mathrm{X} & [0,0] & [0,0] \\
[0,1] & [0,1] & \mathrm{X} & [-1,1] & [-1,1] & \mathrm{X} & [0,0] & [0,0] & [0,0] \\
[0,1] & [0,1] & \mathrm{X} & [-1,1] & \mathrm{X} & [0,0] & [0,0] & [0,0] & [0,0] \\
[0,1] & [0,1] & \mathrm{X} & \mathrm{X} & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] \\
\mathrm{X} & \mathrm{X} & \mathrm{X} & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] \\
[0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0]
\end{pmatrix}$$

```
(%i31)
       /* (1f) */
       [xmin,xmax, ymin,ymax] : [0,9, 0,7];
```
(%o31)
$$[0,9,0,7]$$

```
(%i32) Q(t) := [0,2] + t*[1,1];
```
(%o32)
$$Q\left(t\right) := [0,2]+t\,[1,1]$$

```
(%i33) define(xQ(t), Q(t)[1]);
```
(%o33)
$$\mathrm{xQ}\left(t\right) := t$$

```
(%i34) define(yQ(t), Q(t)[2]);
```
(%o34)
$$\mathrm{yQ}\left(t\right) := t+2$$

```
(%i35)                    [x=xQ(t),x=yQ(t)];
```
(%o35)
$$[x=t,x=t+2]$$

Questão 1: gabarito (5)

```
(%i36) define(h(t), at(z, [x=xQ(t),y=yQ(t)]));
(%o36)
```

$$h(t) := \min(6, \max(0, \min(\max(3, \min(t-2, 2\ (t+2)-12)), t+1)))$$

```
(%i37) myqdrawp(xyrange(), myex1(h(x), lc(red)));
(%o37)
```




```
(%i38)
```

Questão 2: gabarito

```
(%i1) mkmatrix5(x,xs,y,ys,expr) ::=
        buildq([x,xs,y,ys,expr],
               apply('matrix,
                     makelist(makelist(expr,x,xs),y,ys)))$
(%i2)
      /* Algumas definicoes */
      gradef(W  (x,y), W_x (x,y), W_y (x,y))$
(%i3) gradef(W_x(x,y), W_xx(x,y), W_xy(x,y))$
(%i4) gradef(W_y(x,y), W_xy(x,y), W_yy(x,y))$
(%i5) diff6(F)         := [F,
                          diff(F,x),   diff(F,y),
                          diff(F,x,2), diff(F,x,1,y,1), diff(F,y,2)]$
(%i6) M6_(a,b,c,d,e,f) := matrix([a,"",""], [b,c,""], [d,e,f])$
(%i7) T6_(a,b,c,d,e,f) := a + b*Dx + c*Dy + d*Dx^2/2 + e*Dx*Dy + f*Dy^2$
(%i8) M6 (abcdef)      := apply('M6_, abcdef)$
(%i9) T6 (abcdef)      := apply('T6_, abcdef)$
(%i10) atxy(expr,x0y0)  := at(expr, [x=x0y0[1], y=x0y0[2]])$
(%i11) D2(F)            := M6(diff6(F))$
(%i12) T2(x0y0,F)       := T6(atxy(diff6(F),x0y0))$
(%i13) DxDyat(x0y0)     := [Dx=x-x0y0[1], Dy=y-x0y0[2]]$
(%i14) T2exp(x0y0,F)    := subst(DxDyat(x0y0), T2(x0y0,F))$

(%i15)
       M6_ (1,2,3,4,5,6);
```
(%o15)

$$\begin{pmatrix} 1 & & \\ 2 & 3 & \\ 4 & 5 & 6 \end{pmatrix}$$

```
(%i16) M6 ([1,2,3,4,5,6]);
```
(%o16)

$$\begin{pmatrix} 1 & & \\ 2 & 3 & \\ 4 & 5 & 6 \end{pmatrix}$$

```
(%i17)          D2(W(x,y));
```
(%o17)

$$\begin{pmatrix} W(x,y) & & \\ \mathrm{W\_x}(x,y) & \mathrm{W\_y}(x,y) & \\ \mathrm{W\_xx}(x,y) & \mathrm{W\_xy}(x,y) & \mathrm{W\_yy}(x,y) \end{pmatrix}$$

```
(%i18)      diff6(W(x,y));
```
(%o18)

$$[W(x,y), \mathrm{W\_x}(x,y), \mathrm{W\_y}(x,y), \mathrm{W\_xx}(x,y), \mathrm{W\_xy}(x,y), \mathrm{W\_yy}(x,y)]$$

```
(%i19) atxy(diff6(W(x,y)),[x0,y0]);
```
(%o19)

$$[W(\mathrm{x0},\mathrm{y0}), \mathrm{W\_x}(\mathrm{x0},\mathrm{y0}), \mathrm{W\_y}(\mathrm{x0},\mathrm{y0}), \mathrm{W\_xx}(\mathrm{x0},\mathrm{y0}), \mathrm{W\_xy}(\mathrm{x0},\mathrm{y0}), \mathrm{W\_yy}(\mathrm{x0},\mathrm{y0})]$$

```
(%i20) DxDyat([3,4]);
```
(%o20)

$$[\mathrm{Dx} = x - 3, \mathrm{Dy} = y - 4]$$

```
(%i21) T2    ([3,4],W(x,y));
```
(%o21)

$$\mathrm{W\_yy}(3,4)\,\mathrm{Dy}^2 + \mathrm{W\_xy}(3,4)\,\mathrm{Dx}\,\mathrm{Dy} + \mathrm{W\_y}(3,4)\,\mathrm{Dy} + \frac{\mathrm{W\_xx}(3,4)\,\mathrm{Dx}^2}{2} + \mathrm{W\_x}(3,4)\,\mathrm{Dx} + W(3,4)$$

```
(%i22) T2exp ([3,4],W(x,y));
```
(%o22)

$$\mathrm{W\_xy}(3,4)\,(x-3)\,(y-4) + \mathrm{W\_y}(3,4)\,(y-4) + \mathrm{W\_yy}(3,4)\,(y-4)^2 + \mathrm{W\_x}(3,4)\,(x-3) + \frac{\mathrm{W\_xx}(3,4)\,(x-3)^2}{2} + W(3,4)$$

Questão 2: gabarito (2)

```
(%i23) F  : x*y*(6 -2*x -y);
```
(%o23)
$$x\,(-y-2\,x+6)\,y$$

```
(%i24) F  : expand(F);
```
(%o24)
$$-\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y$$

```
(%i25) P1 : [0,6]$
(%i26) P2 : [1,2]$
(%i27) P3 : [3,0]$
(%i28) P4 : [0,0]$
(%i29)
       /* (2a) */
       D2F   :      D2(F);
```
(%o29)
$$\begin{pmatrix} -\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y & & \\ -y^2-4\,x\,y+6\,y & -\left(2\,x\,y\right)-2\,x^2+6\,x & \\ -\left(4\,y\right) & -\left(2\,y\right)-4\,x+6 & -\left(2\,x\right) \end{pmatrix}$$

```
(%i30)
       /* (2b) */
       D2FP1 : atxy(D2(F),P1);
```
(%o30)
$$\begin{pmatrix} 0 & & \\ 0 & 0 & \\ -24 & -6 & 0 \end{pmatrix}$$

```
(%i31) D2FP2 : atxy(D2(F),P2);
```
(%o31)
$$\begin{pmatrix} 4 & & \\ 0 & 0 & \\ -8 & -2 & -2 \end{pmatrix}$$

```
(%i32) D2FP3 : atxy(D2(F),P3);
```
(%o32)
$$\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & -6 & -6 \end{pmatrix}$$

```
(%i33) D2FP4 : atxy(D2(F),P4);
```
(%o33)
$$\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & 6 & 0 \end{pmatrix}$$

```
(%i34) /* (2c) */
       T2(P1,F);
```
(%o34)
$$-\left(6\,\mathrm{Dx}\,\mathrm{Dy}\right)-12\,\mathrm{Dx}^2$$

```
(%i35) T2(P2,F);
```
(%o35)
$$-\left(2\,\mathrm{Dy}^2\right)-2\,\mathrm{Dx}\,\mathrm{Dy}-4\,\mathrm{Dx}^2+4$$

```
(%i36) T2(P3,F);
```
(%o36)
$$-\left(6\,\mathrm{Dy}^2\right)-6\,\mathrm{Dx}\,\mathrm{Dy}$$

```
(%i37) T2(P4,F);
```
(%o37)
$$6\,\mathrm{Dx}\,\mathrm{Dy}$$

```
(%i38)
       /* (2d) */
       grad(F) := [diff(F,x),diff(F,y)]$
(%i39)    H(F) := hessian(F, [x,y])$
(%i40) detH(F) := determinant(H(F))$
(%i41) crit(F) := [F, grad(F), H(F), detH(F)]$

(%i42) crit(W(x,y));
```
(%o42)
$$\left[W\,(x,y)\,,\left[\mathrm{W\_x}\,(x,y)\,,\mathrm{W\_y}\,(x,y)\right],\begin{pmatrix} \mathrm{W\_xx}\,(x,y) & \mathrm{W\_xy}\,(x,y) \\ \mathrm{W\_xy}\,(x,y) & \mathrm{W\_yy}\,(x,y) \end{pmatrix},\mathrm{W\_xx}\,(x,y)\;\mathrm{W\_yy}\,(x,y)-\mathrm{W\_xy}\,(x,y)^2\right]$$

```
(%i43) atxy(crit(F), P1);
```
(%o43)
$$\left[0,[0,0]\,,\begin{pmatrix} -24 & -6 \\ -6 & 0 \end{pmatrix},-36\right]$$

```
(%i44) atxy(crit(F), P2);
```
(%o44)
$$\left[4,[0,0]\,,\begin{pmatrix} -8 & -2 \\ -2 & -2 \end{pmatrix},12\right]$$

```
(%i45) atxy(crit(F), P3);
```
(%o45)
$$\left[0,[0,0]\,,\begin{pmatrix} 0 & -6 \\ -6 & -6 \end{pmatrix},-36\right]$$

```
(%i46) atxy(crit(F), P4);
```
(%o46)
$$\left[0,[0,0]\,,\begin{pmatrix} 0 & 6 \\ 6 & 0 \end{pmatrix},-36\right]$$

Questão 2: gabarito (3)

(%i47)
```
/* (2e), preparacao */
F;
```
(%o47)
$$-\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y$$

(%i48) `T2   (P2,F);`

(%o48)
$$-\left(2\,\mathrm{Dy}^2\right)-2\,\mathrm{Dx}\,\mathrm{Dy}-4\,\mathrm{Dx}^2+4$$

(%i49) `G : T2exp(P2,F);`

(%o49)
$$-\left(2\,\left(x-1\right)\,\left(y-2\right)\right)-2\,\left(y-2\right)^2-4\,\left(x-1\right)^2+4$$

(%i50) `G : expand(G);`

(%o50)
$$-\left(2\,y^2\right)-2\,x\,y+10\,y-4\,x^2+12\,x-12$$

(%i51)
```
atxy(D2(F),P2);
```
(%o51)
$$\begin{pmatrix} 4 & & \\ 0 & 0 & \\ -8 & -2 & -2 \end{pmatrix}$$

(%i52) `atxy(D2(G),P2);`

(%o52)
$$\begin{pmatrix} 4 & & \\ 0 & 0 & \\ -8 & -2 & -4 \end{pmatrix}$$

(%i53)
```
numsB(expr) :=
  apply(matrix,
        makelist(makelist(ev(expr), x,0,3),
                    y, seqby(6,0,-1)))$
```

(%i54) `numsB([x,y]);`

(%o54)
$$\begin{pmatrix} [0,6] & [1,6] & [2,6] & [3,6] \\ [0,5] & [1,5] & [2,5] & [3,5] \\ [0,4] & [1,4] & [2,4] & [3,4] \\ [0,3] & [1,3] & [2,3] & [3,3] \\ [0,2] & [1,2] & [2,2] & [3,2] \\ [0,1] & [1,1] & [2,1] & [3,1] \\ [0,0] & [1,0] & [2,0] & [3,0] \end{pmatrix}$$

(%i55)
```
/* (2e) */
numsB(F);
```
(%o55)
$$\begin{pmatrix} 0 & -12 & -48 & -108 \\ 0 & -5 & -30 & -75 \\ 0 & 0 & -16 & -48 \\ 0 & 3 & -6 & -27 \\ 0 & 4 & 0 & -12 \\ 0 & 3 & 2 & -3 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

(%i56)
```
/* (2f) */
numsB(G);
```
(%o56)
$$\begin{pmatrix} -24 & -28 & -40 & -60 \\ -12 & -14 & -24 & -42 \\ -4 & -4 & -12 & -28 \\ 0 & 2 & -4 & -18 \\ 0 & 4 & 0 & -12 \\ -4 & 2 & 0 & -10 \\ -12 & -4 & -4 & -12 \end{pmatrix}$$

(%i57)
```
/* (2g) */
[xmin,ymin, xmax,ymax] : [-1,-1, 4,7]$
```

```
(%i58) level(zz,color) := myimp1(F=zz, lc(color), lk(z=zz))$
(%i59) levels() := [level( 3.95, gray),
                    level( 3.90, gray),
                    level( 3.85, gray),
                    level( 3,    red),
                    level( 2,    orange),
                    level( 1,    gold),
                    level( 0,    forest_green),
                    level(-1,    blue)]$
```

Questão 2: gabarito (3)

```
(%i60)
       /* As curvas de nivel da F, sem os vetores gradientes: */
       level(zz,color) := myimp1(F=zz, lc(color), lk(z=zz))$

(%i61) myqdrawp(xyrange(), levels());
(%o61)
```




```
(%i62)
       /* As curvas de nivel da G, sem os vetores gradientes: */
       level(zz,color) := myimp1(G=zz, lc(color), lk(z=zz))$

(%i63) myqdrawp(xyrange(), levels());
(%o63)
```




```
(%i64) /* Os conjuntos B e C: */
       B : create_list([x,y], y,seqby(6,0,-1), x,seq(0,3));
(%o64)
```

$[[0,6],[1,6],[2,6],[3,6],[0,5],[1,5],[2,5],[3,5],[0,4],[1,4],[2,4],[3,4],[0,3],[1,3],[2,3],[3,3],[0,2],[1,2],[2,2],[3,2],[0,1],[1,1],[2,1],[3,1],[0,0],[1,0],[2,0],[3,0]]$

```
(%i65) eq1 : y = 6 - 2*x;
(%o65)
```

$$y = 6 - 2x$$

```
(%i66) eq2 : solve(eq1,x);
(%o66)
```

$$\left[x = -\left(\frac{y-6}{2}\right)\right]$$

```
(%i67) subst(eq2, x);
(%o67)
```

$$-\left(\frac{y-6}{2}\right)$$

```
(%i68) define(xmaxC(y), subst(eq2, x));
(%o68)
```

$$\mathrm{xmaxC}\,(y) := -\left(\frac{y-6}{2}\right)$$

```
(%i69) C : create_list([x,y], y,seqby(6,0,-1), x,seq(0,xmaxC(y)));
(%o69)
```

$[[0,6],[0,5],[0,4],[1,4],[0,3],[1,3],[0,2],[1,2],[2,2],[0,1],[1,1],[2,1],[0,0],[1,0],[2,0],[3,0]]$

```
(%i70)
       [myqdrawp(xyrange(), pts(B, pc(red), myps(3))),
        myqdrawp(xyrange(), pts(C, pc(red), myps(3)))];
(%o70)
```





Questão 2: gabarito (4)

```
(%i71) /* O gradiente da F nos pontos de B: */
       numsB(     grad(F)       );
(%o71)
```

$$\begin{pmatrix} [0,0] & [-24,-8] & [-48,-20] & [-72,-36] \\ [5,0] & [-15,-6] & [-35,-16] & [-55,-30] \\ [8,0] & [-8,-4] & [-24,-12] & [-40,-24] \\ [9,0] & [-3,-2] & [-15,-8] & [-27,-18] \\ [8,0] & [0,0] & [-8,-4] & [-16,-12] \\ [5,0] & [1,2] & [-3,0] & [-7,-6] \\ [0,0] & [0,4] & [0,4] & [0,0] \end{pmatrix}$$

```
(%i72) numsB(atxy(grad(F),[x,y]));
(%o72)
```

$$\begin{pmatrix} [0,0] & [-24,-8] & [-48,-20] & [-72,-36] \\ [5,0] & [-15,-6] & [-35,-16] & [-55,-30] \\ [8,0] & [-8,-4] & [-24,-12] & [-40,-24] \\ [9,0] & [-3,-2] & [-15,-8] & [-27,-18] \\ [8,0] & [0,0] & [-8,-4] & [-16,-12] \\ [5,0] & [1,2] & [-3,0] & [-7,-6] \\ [0,0] & [0,4] & [0,4] & [0,0] \end{pmatrix}$$

```
(%i73)
       /* O gradiente da F nos pontos de C: */
       define(v_at (xy), atxy(grad(F)/10,xy))$

(%i74) define(Pv_at(xy), myPv(xy,v_at(xy),[ps(1)],hl(0.15)))$

(%i75) Pv_at([1,3]);
(%o75)
```

$$\left[\mathrm{pts}\left([[1,3]],\mathrm{ps}(1)\right),\mathrm{vector}\left([1,3],\left[-\left(\frac{3}{10}\right),-\left(\frac{1}{5}\right)\right],\mathrm{hl}(0.15)\right)\right]$$

```
(%i76) grads_at_C : makelist(Pv_at(xy), xy, C)$

(%i77) myqdrawp(xyrange(), grads_at_C);
(%o77)
```




```
(%i78)
       /* O gradiente da F nos pontos de C e as curvas de nivel: */
       level(zz,color) := myimp1(F=zz, lc(color), lk(z=zz))$

(%i79) myqdrawp(xyrange(), levels(), grads_at_C);
(%o79)
```




```
(%i80)
```